A határozott integrál alkalmazása

Területszámítás

Definíció

Legyen f az (a,b) intervallumon előjelet nem váltó, korlátos és legfeljebb véges számú ponttól eltekintve folytonos valós függvény(az $a = -\infty$ és a $b = \infty$ esetet is megengedve).
A síkbeli derékszögű koordináta-rendszerben

az $y = f(x)$ egyenletű görbe
az (a,b) intervallum, valamint
az x = a és az x = b egyenletű egyenes

által határolt síkidomot az f függvény és az (a,b) intervallum által meghatározott **görbevonalú trapéz**nak nevezzük. Ha $a = -\infty$,akkor a nem létező x = a egyenes, ha pedig $b = \infty$, akkor a nem létező x = b egyenletű egyenes figyelmen kívül hagyandó.
(A definíciónak az előjelre vonatkozó követelménye úgy is megfogalmazható, hogy az (a,b) intervallumon mindvégig $0 \le f(x)$, vagy mindvégig $f(x) \le 0$ legyen)
Véges (a,b) intervallum esetén a megfelelő görbevonalú trapéz területére érvényes a következő:

Tétel

Ha az egyváltozós valós f függvény a véges (a,b) intervallummal együtt görbevonalú trapézt határoz meg, akkor ezen a görbevonalú trapéz a területének a számértéke az

$$\int_a^b |f(x)|\, dx$$

határozott integrállal egyenlő.

Megjegyzés: Ha az f függvény a figyelembe veendő (a,b) intervallum egyes részintervalumain az $0 \le f(x)$, az (a,b) többi részintervallumán pedig az $f(x) \le 0$ egyenlőtlenségnek tesz eleget, akkor részintervallumonként számítjuk ki a területet, és az ezekre kapott értékeket összeadjuk.

Példa

Határozzuk meg az $f(x) = \sin(x) + \sin(2x) + \sin(3x)$ *függvény görbe alatti területét a* $[0, 2\pi]$ *intervallumon!*

Megoldás:

Vegyük fel, s ábrázoljuk a függvényt!

```
> f:=x->sin(x)+sin(2*x)+sin(3*x);
```

$$f := x \to \sin(x) + \sin(2x) + \sin(3x) \quad \textbf{(1.1)}$$

```
> plot(f(x),x=0..2*Pi);
```



A grafikon szemmel láthatóan szimmetrikus a (π, 0) pontra. Ezt könnyen igazolhatjuk is! A függvényértékek a π-re szimmetrikus helyeken:

```
> f(Pi-x);
```

$$\sin(x) - \sin(2x) + \sin(3x) \tag{1.2}$$

```
> f(Pi+x);
```

$$-\sin(x) + \sin(2x) - \sin(3x) \tag{1.3}$$

Az is utasítás is igaz értéket ad:

```
> is(f(Pi-x)=-f(Pi+x));
```

$$\mathit{true} \tag{1.4}$$

Így a [0,π] számközre eső érték kétszerese a keresett terület:

```
> T:=2*(Int(f(x),x=0..Pi/2)-Int(f(x),x=Pi/2..2*Pi/3)+Int(f(x),
x=2*Pi/3..Pi));
```

$$T := 2\left(\int_0^{\frac{1}{2}\pi} \left(\sin(x) + \sin(2x) + \sin(3x)\right)\,dx\right) - 2\left(\int_{\frac{1}{2}\pi}^{\frac{2}{3}\pi} \left(\sin(x) + \sin(2x) + \sin(3x)\right)\right. \tag{1.5}$$

$$\left. dx \right) + 2 \left(\int_{\frac{2}{3}\pi}^{\pi} (\sin(x) + \sin(2x) + \sin(3x))\, dx \right)$$

```
> `A görbe alatti terület`=value(T);
```

$$A\ görbe\ alatti\ terület = \frac{17}{3} \qquad (1.6)$$

Görbék közötti terület

Tétel

Legyen $f(a) = g(a)$, $f(b) = g(b)$ *és* $g(x) < f(x)$ *minden (a,b)-beli x-re. Ekkor a két görbe által bezárt terület mérőszáma:*

$$T = \int_a^b (f(x) - g(x))\, dx$$

Tekintsük a következő két függvényt:

```
> f:=x->-(x-1)^2+3;
```

$$f := x \to -(x-1)^2 + 3 \qquad (1.7)$$

```
> g:=x->x^2-2;
```

$$g := x \to x^2 - 2 \qquad (1.8)$$

```
> plot([f(x),g(x)],x=-1.41..2.1,thickness=2,title=`Görbék közti terület`):
>
```

Határozzuk meg a két görbe metszésgontjainak abszcissszáját:

```
> solve(f(x)=g(x),x);
```

$$-1, 2 \qquad (1.9)$$

Ha mindkét görbét az y tengely mentén pozitív irányban eltoljuk annyival, hogy a kisebb értékeket felvevő is pozitívvá váljon, akkor látható, hogy a két terület különbsége adja a görbék közötti területet, tehát a képlet ilyenkor is érvényes. (az eltolás mértékét adó C "kiesik")

```
> with(plots): with(plottools):
> f:=x->-(x-1)^2+3+2.5;
```

$$f := x \to -(x-1)^2 + 3 + 2.5 \qquad (1.10)$$

```
> g:=x->x^2-2+2.5;
```

$$g := x \to x^2 - 2 + 2.5 \qquad (1.11)$$

```
> abra:=plot([f(x),g(x)],x=-1.41..2.1,thickness=2):
> vonal1:=line([-1,0],[-1,f(-1)], color=yellow, linestyle=3):
> vonal2:=line([2,0],[2,f(2)], color=yellow, linestyle=3):
> display([abra,vonal1,vonal2],title=`A két görbét eltoljuk úgy, hogy a minimum pozitív legyen`):
>
```

```
> T:=Int(simplify(f(x)-g(x)),x=-1..2);
>
```

$$T := \int_{-1}^{2} \left(-2.\,x^2 + 2.\,x + 4. \right) \mathrm{d}x \qquad (1.12)$$

```
> `A görbék közti terület`=value(T);
```

$$A\ görbék\ közti\ terület = 9. \qquad (1.13)$$

Paraméteres alakban adott görbék területszámítása.

Legyen $x = f(t)$;

$y = g(t)$; $\alpha \le t <= \beta$

egy görbe paraméteres egyenletrendszere, és legyen

$a = f(\alpha)$, $b = f(\beta)$,

ekkor az [a,b] intervallumhoz tartozó görbevonalú trapéz területe:

$$T = \int_{\alpha}^{\beta} g(t) \left(\frac{\mathrm{d}}{\mathrm{d}t} f(t) \right) \mathrm{d}t$$

A képlet akkor is érvényes, ha a fenti paraméteres egyenletrendszerrel adott görbe zárt, ilyenkor a zárt görbe által határolt síkidom területét kapjuk meg.

Példa

Számítsuk ki az

$$x = \cos(t)^3;\ y = \sin(t)^3 \quad 0 <= t \le 2\,\pi$$

egyenletrendszerrel adott asztroida területét!

Megoldás: Ábrázoljuk a görbét:

```
> plot([ cos(t)^3, sin(t)^3, t=0..2*Pi], thickness=2, color=
  magenta, scaling=constrained);
```




```
> f:=t->cos(t)^3;
```

$$f := t \rightarrow \cos(t)^3 \quad \textbf{(1.14)}$$

```
> g:=t->sin(t)^3;
```

$$g := t \rightarrow \sin(t)^3 \quad \textbf{(1.15)}$$

```
> T:=Int(g(t)*diff(f(t),t),t=0..2*Pi);
```

$$T := \int_0^{2\pi} \left(-3 \sin(t)^4 \cos(t)^2\right) \mathrm{d}t \quad \textbf{(1.16)}$$

```
> `Az asztroida területe`=value(T);
```

$$\mathit{Az\ asztroida\ területe} = -\frac{3}{8}\pi \quad \textbf{(1.17)}$$

A negatív előjel itt az óramutató járásával ellentétes irányú körülfutás miatt van.

Térfogatszámítás

A forgástest térfogata

A forgástest térfogatának meghatározásánál a következő két szemléletes elvből indulunk ki:

1.A testek térfogata additív, azaz az egész test térfogata egyenlő a részek térfogatának összegével;

2. Az egyenes körhenger térfogata egyenlő a henger alapterületének és a magasságának a szorzatával.
Ezek után legyen $y = f(x)$ (>=0) egy, az [a,b] zárt intervallumban folytonos függvénygörbe, amelyet az x tengely körül megforgatunk. Határozzuk meg annak a forgástestnek a térfogatát, amelyet az [a,b] számközhöz tartozó görbeív és a két szélső ordinátaszakasz megforgatásával kapunk!
Legyen a = $x_0 < x_1 < x_2 < \ldots < x_n$ = b az [a,b] intervallum egy tetszőleges beosztása, és képezzük az ezen beosztáshoz tartozó beírt és körülírt lépcsős sokszöget.
Például:

```
> f:=x->x^2+1;
```

$$f := x \rightarrow x^2 + 1 \qquad \textbf{(2.1.1)}$$

```
> abra1:=student[leftbox](f(x),x=0..2,shading=YELLOW):
> abra2:=student[rightbox](f(x),x=0..2,shading=BLUE):
> plots[display](abra1,abra2,title=`A lépcsős sokszögek`);
```

A lépcsős sokszögek



Ha ezeket a sokszögeket a görbével együtt forgatjuk, akkor ezek is leírnak egy-egyforgástestet, amelyek térfogata közrefogja a görbe által leírt forgástest térfogatát.
A lépcsős sokszögek körülforgatásakor a lépcsős sokszög téglalapjai egy-egy hengert írnak le.
Ha az f(x) függvénynek az $[x_{k-1}, x_k]$ intervallumo felvett legkisebb értéke m_k, legnagyobb értéke pedig M_k, akkor a k-adik részintervallumhoz tartozó beírt, illetve körülírt henger térfogata

$$\pi\, m_k^2\,(x_k - x_{k-1}),\quad \text{illetve}\quad \pi\, M_k^2\,(x_k - x_{k-1})$$

(k =1,2, . . .,n), és a görbe körülforgatásából származó forgástest V térfogata a beírt és körülírt hengerek térfogatának összege közé esik:

$$\sum_{k=1}^{n} \pi\, m_k^2\,(x_k - x_{k-1}) <= V \le \sum_{k=1}^{n} \pi\, M_k^2\,(x_k - x_{k-1}).$$

Minthogy feltevésünk szerint $0 \le f(x)$(ez a forgástestre nézve nyilvánvalóan nem jelent megkötést), ezért $\pi\, m_k^2$ a $\pi f(x)^2$ függvénynek az $[x_{k-1}, x_k]$ intervallumon felvett legkisebb, $\pi\, M_k^2$ pedig a legnagyobb értéke(k = 1,2,...,n). Így

$$\sum_{k=1}^{n} \pi\, m_k^2\,(x_k - x_{k-1}),\ \text{ill.}\quad \sum_{k=1}^{n} \pi\, M_k^2\,(x_k - x_{k-1})$$

a $\pi f(x)^2$ függvénynek az [a,b] intervallum tekintett beosztásához tartozó alsó, ill. felső integrálközelítő összege. Ha a beosztás az [a,b] minden határon túl finomodó beosztássorozatán fut át, akkor ezek konvergálnak az

$$\int_a^b \pi f(x)^2\, dx$$

integrálhoz. A fenti egyenlőtlenségből a közrefogási szabály alapján következik, hogy

$$\mathbf{V} = \int_a^b \pi f(x)^2\, dx = \pi \int_a^b f(x)^2\, dx.$$

Példa

Számítsuk ki az ***f(x) = sin(x)*** *[0,π] intervallumra eső ívének az x tengely körüli forgatásakor keletkező* forgástest térfogatát*!*

```
> V:=Int(Pi*sin(x)^2,x=0..Pi);
```

$$V := \int_0^{\pi} \pi \sin(x)^2\, dx \tag{2.1.2}$$

```
> `A térfogat`=value(V);
```

$$\mathit{A\ térfogat} = \frac{1}{2}\pi^2 \tag{2.1.3}$$

Ha a körülforgatott görbe egyenlete paraméteres alakban van megadva:

$x = f(t)$;

$y = g(t)$; ($\alpha \le t <= \beta$).

akkor a térfogat

$$V = \pi \left(\int_{\alpha}^{\beta} g(t)^2 \left(\frac{d}{dt} f(t) \right) dt \right).$$

Példa

Számítsuk ki az $x = r\,(t - \sin(t))$; $y = r\,(1 - \cos(t))$ (0<=t<=2 π) cikloisnak az x tengely körüli forgatásakor keletkező forgástest térfogatát!

Megoldás

Mindenekelőtt ábrázoljuk magát a görbét, majd a forgástestet!

```
> r:=4;
```

$$r := 4 \tag{2.1.4}$$

```
> plot([r*(t-sin(t)),r*(1-cos(t)),t=0..2*Pi],scaling=constrained,color=green,thickness=2,title=`a ciklois`);
```

a ciklois




```
> plot3d([r*(t-sin(t)),r*(1-cos(t))*sin(v),r*(1-cos(t))*cos(v)],t=0..2*Pi,v=0..2*Pi,grid=[40,40],title=`A cikloid`);
```



A képlet alapján:

```
> V: =Pi * I nt ( r * ( 1- cos( t ) ) ^2* di f f ( r * ( t - si n( t ) ) , t ) , t =0. . 2* Pi ) ;
>
```

$$V := \pi \left(\int_0^{2\pi} 4\ (1 - \cos(t))^2\ (4 - 4\cos(t))\ \mathrm{d}t \right) \tag{2.1.5}$$

```
> V: =si mpl i f y( Pi * I nt ( ( r * ( 1- cos( t ) ) ) ^2* di f f ( r * ( t - si n( t ) ) , t ) , t =
  0. . 2* Pi ) ) ;
```

$$V := 64\,\pi \left(\int_0^{2\pi} \left(1 - \cos(t)^3 + 3\cos(t)^2 - 3\cos(t)\right)\ \mathrm{d}t \right) \tag{2.1.6}$$

```
> `A ci kl oi d t ér f ogat a` =val ue( V) ;
```

$$\textit{A cikloid térfogata} = 320\,\pi^2 \tag{2.1.7}$$

Eljárás a forgástest térfogatára

A térfogat eljárás kiszámítja az $f(x)$ függvény $[a, b]$ intervallumra eső ívének az x tengely körüli megforgatásakor keletkező forgásfelület és a két megfelelő körlemez által határolt forgástest térfogatának számértékét és szemlélteti a forgásfelületet.

```
> térfogat:=proc()
local  f,a,b,vol,pont,gráf:
f:=args[1]:
a:=args[2]:b:=args[3]:
vol:=value(Int(Pi*f(x)^2,x=a..b));
if  not(type(vol/Pi,numeric))  then
vol:=abs(evalf(vol))
fi:
print('V'=vol);
if  type(vol,numeric)  or  type(vol,realcons)  then
plot3d([t,f(t)*cos(v),f(t)*sin(v)],t=a..b,v=0..2*Pi,style=
patch,lightmodel=light3,orientation=[-64,52],axes=normal);
fi;
end proc:
```

```
> f:=x->exp(x/2)*sin(x);
```

$$f := x \rightarrow e^{\frac{1}{2}x} \sin(x) \quad \textbf{(2.2.1)}$$

```
> térfogat(f,0,5);
```

$$V = 321.7225979$$



>

A síkgörbe ívhossza

Értelmezés

Egy görbe ívhosszának azt a határértéket nevezzük, amelyhez a beírt törtvonal hossza tart, ha oldalainak hossza minden határon túl nő, úgy, hogy közben a leghosszabb oldal hossza is 0-hoz tart. Az olyan görbéket, amelyeknek van ívhosszuk, **rektifikálható görbék**nek nevezzük

Tétel

Ha az f(x) függvény f ' (x) differenciálhányadosa folytonos az [a,b] zárt intervallumon, akkor grafikonjának az [a,b] intervallumhoz tartozó íve rektifikálható, és ezen ív hossza:

$$L = \int_a^b \sqrt{1 + \left(\frac{d}{dx} f(x)\right)^2}\, dx.$$

Példa

Számítsuk ki az f(x) $= x^2$ parabola [0,1] intervallumhoz tartozó ívének hosszát!

```
> with(student):
> f:=x->x^2;
```

$$f := x \rightarrow x^2 \tag{3.1}$$

```
> L:=Int(sqrt(1+diff(x^2,x)^2),x=0..1);
```

$$L := \int_0^1 \sqrt{1 + 4x^2}\, dx \tag{3.2}$$

```
> `Az ívhossz értéke`=value(L);
```

$$\textit{Az ívhossz értéke} = \frac{1}{2}\sqrt{5} - \frac{1}{4}\ln\left(-2 + \sqrt{5}\right) \tag{3.3}$$

A paraméteres alakban adott görbék ívhossza:
Ha a görbe paraméteres egyenletrendszere az előbbi, akkor ívhosszát az

$$L = \int_{\alpha}^{\beta} \sqrt{\left(\frac{d}{dt} f(t)\right)^2 + \left(\frac{d}{dt} g(t)\right)^2}\, dt$$

képlet alapján számíthatjuk ki.

A forgástest palástjának felszíne

Tétel

.Ha az f függvény differenciálhányadosa az [a,b] intervallumon folytonos, akkor f(x) [a,b]-hez tartozó ívének az x tengely körüli forgatásával előálló forgásfelület felszíne:

$$F = 2\pi \left(\int_a^b f(x) \sqrt{1 + \left(\frac{d}{dx} f(x)\right)^2}\, dx \right).$$

Ha a körülforgatott görbe paraméteres egyenletrendszere

$$x = f(t);$$

$$y = g(t);\ (\alpha \le t \le \beta),$$

akkor a görbe $\alpha \le t \le \beta$ íve által meghatározott forgásfelület felszíne:

$$F = 2\pi \left(\int_{\alpha}^{\beta} g(t) \sqrt{\left(\frac{d}{dt} f(t)\right)^2 + \left(\frac{d}{dt} g(t)\right)^2}\, dt \right)$$

Példa

Határozzuk meg a cikloisz egy ívének az x tengely körüli megforgatásával előálló forgásfelület felszínét!

Ha a cikloisz egyenletrendszere:

$$x = r\,(t - \sin(t));\ y = r\,(1 - \cos(t)),\ (0 \le t \le 2\pi),$$

akkor a forgástest palástjának felszíne az utóbbi képlettel:

```
> r:='r':
```

```
> F:=2*Pi*Int(r*(1-cos(t))*sqrt(r^2*(1-cos(t))^2+r^2*(sin(t)^2)),t=0..2*Pi);
```

$$F := 2\,\pi \left(\int_{0}^{2\,\pi} r\,(1-\cos(t))\,\sqrt{r^2\,(1-\cos(t))^2 + r^2\,\sin(t)^2}\,\mathrm{d}t \right) \tag{4.1}$$

```
> `A palást felszíne`=simplify(value(F));
```

$$\textit{A palást felszíne} = \frac{64}{3}\,\pi\,r^2\,\mathrm{csgn}(r) \tag{4.2}$$

```
> assume(r>0):
> `A palást felszíne`=simplify(value(F));
```

$$\textit{A palást felszíne} = \frac{64}{3}\,\pi\,r\sim^2 \tag{4.3}$$

```
>
```